```
// Tyler Stevens & Graydon Hegge - ME461 Final Projecct - Cheadle - Aug 5, 2021

$UnitSystem SI Mass kJ C Pa Degrees
$TabStops 0.2 0.4 0.6 4
$Sumrow on

"isentropic pump procedure"
Procedure pump(h_in,P_in,P_out,F$,eta_p:h_out,T_out,s_out,v_out)
    m=1[kg]
    s_in=entropy(F$,h=h_in,P=P_in)
    s_s_out=s_in
    h_s_out=enthalpy(F$,P=P_out,s=s_s_out)
    W_p_s = m*(h_s_out - h_in)
    W_p = W_p_s/eta_p
    h_out = h_in + W_p/m
    T_out=temperature(F$,h=h_out,P=P_out)
    s_out=entropy(F$,h=h_out,P=P_out)
    v_out=volume(F$,h=h_out,P=P_out)
End

"! SET SYSTEM: Baseline (w/o compost heater) or Modified (w/ compost heater)"
SYS$ = 'base'
//SYS$ = 'mod'

"models - real fluid"
W$ = 'water'
A$ = 'air_ha'
"ambient enviornment"
$IFNOT PARAMETRICTABLE
    T_amb = converttemp(F,C,-20)                      "! DESIGN CONDITION Tamb SET: DC_P==1, F_NG ==1"
    hour = 1
    time = 1
$ENDIF
P_amb = Po#

"water tank"
D_tank = 2*convert(ft,m)
H_tank = 6*convert(ft,m)
SA_tank = (pi*D_tank*H_tank) + 2*(pi*((D_tank/2)^2))
Vol_tank = (pi*(D_tank^2)/4)*H_tank
h_conv = 5 [W/m^2-k]                                  "convection ht coeff from insulation surface"
R_dprime = 8*convert(hr-ft^2-F/Btu,s-m^2-K/J)
R_ins = (R_dprime/SA_tank)*convert(s-K/J,K/W)         "water tank with insulation resistance"
R_conv = 1/(h_conv*SA_tank)
R_tot = R_conv + R_ins
"water tank heat loss"
Q_loss_tank = (((converttemp(C,K,T[2])-converttemp(C,K,T_barn))/R_tot)*3600[s])*convert(J,kJ)

"pump"
eta_p = 0.8                                           "isentropic pump efficiency"
PR = 3[-]                                             "pump pressure ratio"

"floor heat exchanger"
epsilon_h = 0.875                                     "heat exchanger effectiveness"

"composter"
Q_comp_ton = 1000 [BTU/ton]*convert(BTU/ton,kJ/ton)   "heat from comp each hour - per ton"
rho_comp = 640 [kg/m^3]*convert(kg/m^3,ton/m^3)       "density of compost"
Vol_comp = m_comp/rho_comp                            "mass of compost"
F_vol = 0.10                                          "fraction of compost volume to barn volume"
Vol_comp = (Vol_b*convert(m^3,ft^3))*F_vol*convert(ft^3,m^3)   "compost volume scaled to fraction of barn size"
Q_comp = Q_comp_ton*m_comp                            "heat supply per hour from compost"
```

```
"barn"
T_barn = converttemp(F,C,50)                                    "! desired barn temperature setpoint"
L_b = 40*convert(ft,m)
W_b = 30*convert(ft,m)                                          "barn geometry"
H_b = 10*convert(ft,m)
Vol_b = L_b*W_b*H_b
SA_b = 2*H_b*(L_b + W_b)
"equivalent resistance from exterior surface"
R_a = 0.25[hr-ft^2-F/Btu]/(SA_b*convert(m^2,ft^2))
"corregated steel properties"
k_s = 8 [Btu/hr-ft-F]
L_s = 0.0136 [in]*convert(in,ft)
"equivalent conduction resistance through steel walls"
R_s = L_s/k_s/(SA_b*convert(m^2,ft^2))
"equivalend resistance from interior surface"
R_z = 0.68[ft^2-F-hr/Btu]/(SA_b*convert(m^2,ft^2))
DELTA_T_f = converttemp(C,F,T_barn)-converttemp(C,F,T_amb)      "temperature difference between barn and
ambient"
Q_dot_loss = ((DELTA_T_f)/(R_a + R_s+ R_z))
"barn heat loss"
Q_loss_barn = (q_dot_loss*convert(btu/hr,KW))*(60[s/min])*(60[min/hr])*(1[hr])          "actual barn heat loss.
can compare to check value"
Q_loss_barn_check = ((Vol_b*convert(m^3,ft^3))*(0.133[BTU/F-ft^3])*(DELTA_T_f))*convert(BTU,kJ)      "barn heat loss
comparison to eq (1). found in research"
"barn heated under dirt"
SA_d = (L_b*W_b)*convert(m^2,ft^2)
k_d = 0.7*convert(W/m-K,Btu/hr-ft-F)                            "thermal conductivity of soil [0.38-1.7] depending
on saturation"
L_d = 3*convert(in,ft)                                          "heated piped 4in under soil surface"
"equivalent conduction resistance through dirt floor"
R_d = L_d/(k_d*SA_d)
"dirt floor temp determined from heat rate necessary into barn to account for barn loss"
Q_dot_barn = (T_floor - (converttemp(C,F,T_barn)))/R_d
T_barn_f = converttemp(C,F,T_barn)
T_floor_adj = if(T_floor,T_barn_f,T_barn_f,T_barn_f,T_floor)    "adjusted - floor model doesnt get cooled when
hot outside - assume stays at minimum barn temp"

"cattle heat generation"
N_cattle = 20
Q_cattle = 700[kJ]*N_cattle                                     "internal heat generation from livestock"
Q_dot_cattle = Q_cattle*convert(kJ,Btu)/1[hr]

"natural gas heater"
Q_ng = 92333*convert(BTU,kJ)                                    "Natural Gas Heater heat per hour: from design
condition: -20 F"
//F_ng = 1 [-]                                                  "Fraction of ng design condition heat delivered"

"mass flow rate"
m_design = 596.9 [kg]                                           "mass flow water @design cond (-20F)"
DC_p = m/m_design                                               "pump duty cycle - fraction of time on to deliver
mass flow required at design cond"

"piping"
d_pipe = 0.75*convert(in,m)
A_pipe = pi#*(d_pipe/2)^2
"water flow speed check"
u_pipe = (Vol_w/A_pipe)/(3600[s])                               "check for reasonable flow velocity in piping"

"! barn zone balance"
Q_barn + Q_cattle = Q_loss_barn                                 "assume heat to barn + heat from cows == heat
leaving barn"
Q_dot_barn + Q_dot_cattle = Q_dot_loss                          "Solve for Q_barn heat required to maintain barn
temperature"
```

```
"! state 1 - pump outlet"
"state 1 fixed from pump procedure"
P[1] = PR*P[3]
"other calculations"
Vol_w = v[1]*m

"! state 1-2 - tank"
"energy balance"
$IF SYS$ = 'base'
    m*h[1] + (Q_ng*F_ng) = m*h[2] + Q_loss_tank
$ENDIF
$IF SYS$ = 'mod'
    m*h[1] + Q_comp + (Q_ng*F_ng) = m*h[2] + Q_loss_tank
    HF_comp = Q_comp / (Q_comp+(Q_ng*F_ng))                 "heat fraction: compost heat to toal heat
delivered"
$ENDIF

"! state 2 - tank outlet"
T[2] = converttemp(F,C,130)                                 "water outlet temperature setpoint from tank"
P[2] = P[1]
h[2] = enthalpy(W$, T=T[2], P=P[2])
v[2] = volume(W$, T=T[2], P=P[2])
s[2] = entropy(W$, T=T[2], P=P[2])

"! state 2-3 - floor heat exchanger"
"HX performance"
epsilon_h = (T[2]-T[3])/(T[2]-T_barn)                       "solve for T[3] out of HX"
"energy balance"
m*h[2] = m*h[3] + Q_barn

"! state 3 - HX outlet"
P[3] = P_amb
h[3] = enthalpy(W$, T=T[3],P=P[3])
v[3] = volume(W$, T=T[3],P=P[3])
s[3] = entropy(W$,T=T[3],P=P[3])

"! state 3-1 - pump"
Call pump(h[3],P[3],P[1],W$,eta_p:h[1],T[1],s[1],v[1])     "determine state 1 properties, after isentropic
pump"
"energy balance"
m*h[3] + W_p = m*h[1]                                       "electrical work to drive the pump"

"energy used"
E_ng = Q_ng*F_ng_adj
E_elec = W_p*DC_p_adj
E_total = E_ng + E_elec

"cost"
F_ng_adj = if(F_ng,0,0,0,F_ng)                              "set natrual gas heater fraction of heat delivered
to zero to turn off if negative heat required"
DC_p_adj = if(DC_p,0,0,0,DC_p)                              "set pump duty cycle to zero if negative heat
required from comp and ng"
ec = 0.12 [$/kW-hr]*convert($/kw-hr,$/kJ)                   "electricity price"
ngc = 0.95 [$/therm]*convert($/therm,$/kJ)                  "natural gas price"
C_ng = ngc*Q_ng*F_ng_adj                                    "natural gas cost"
C_ec = ec*W_p*DC_p_adj                                      "electricity cost"
COST = C_ng+C_ec                                            "energy costs - not including cost of compost
needed for modified system"

"! SET ==1 TO RUN ECONOMIC ANALYSIS - OTHERWISE SET == 0"
ECONOMICS$ = '1'
$IF ECONOMICS$ = '1'
    "base system winter cost"
    C_winter_base = sumparametric('WINTER-BASE','COST')
```

```
"modified system winter cost"
C_winter_mod = sumparametric('WINTER-MOD','COST')
"cost difference in one winter heating season"
C_saved = C_winter_base - C_winter_mod

"! economics"
N = 10                                          "total time loan"
d = 0.025                                       "market discount rate"
//ROI = d
//ROI = 0.1519                                  "RETURN ON INVESTMENT: market discount
rate, d, when LCS == 0 (@design cond)"
i_tax = 0.25                                    "effective tax rate"
"downpayment and mortgage"
F_comp = 7000 [$]                               "composter total install (capital) cost"
f_dp = 0.15                                     "downpayment fraction"
C_dp = f_dp*F_comp                              "downpayment cost"
Mort = (1-f_dp)*F_comp                          "total mortgage loan amount"
m_rate = 0.03                                   "mortgage interest rate"
"fuel"
cc = 25 [$/ton]                                 "price of materials necessary for compost pile"
cf = 55 [$/ton]                                 "resale price per ton of decomposed fertilizer and
materials at end of each season"
labor = 23 [$/ton]                              "labor price to install compost each year"
i_elec = 0.03                                   "electricity cost inflation rate"
i_comp = 0.03                                   "compost price inflation rate"
i_fert = 0.04                                   "fertilzer price inflation rate"

"initial compost savings"
CS_comp[0] = -F_comp*f_dp
PW_comp[0] = CS_comp[0]                         "present worth of compost savings"
"first year fuel/energy costs"
F_purchase_comp = 0.5                           "fraction of compost that must be purchased
each year"
F_compfuel = cc*m_comp*F_purchase_comp          "compost costs first year"
F_energy_save = C_winter_base - (C_winter_mod + F_compfuel)    "fuel/energy savings between base and modified
systems (accounting for cost of compost materials needed each year)"
"initial compost install labor costs"
CS_labor[0] = -(labor*m_comp)
"annual mortgage payment"
PR[0] = Mort
PR[0] = C_mortgage*pwf(N,0,m_rate)              "determine mortgage cost"

"yearly savings"
   Duplicate j=1,N
      period[j] = j
      "energy savings costs"
      CS_energy[j] = (F_energy_save)*((1+i_elec)^(j-1))      "energy savings at year j"
      "resale"
      CS_resale[j] = (cf*m_comp)*((1+i_fert)^(j-1))          "resale price of fertilizer at end of each year"
      "compost install labor costs"
      CS_labor[j] = -((labor*m_comp)*((1+d)^(j-1)))          "cost of labor at year j"
      "interest and principal"
      C_int[j] = m_rate*PR[j-1]                              "interest at year j"
      PR[j] = PR[j-1] - (C_mortgage - C_int[j])              "principal at year j"
      "income tax"
      CS_tax[j]=C_int[j]*i_tax                               "income tax savings at year j"
      "composter savings"
      CS_comp[j] = CS_energy[j] - C_mortgage + CS_tax[j] + CS_labor[j] + CS_resale[j]      "yearly system savings"
      PW_comp[j] = CS_comp[j]/(1+d)^j                        "respective present worth"
      "cumulative energy savings"
      CS_energy_sum[j] = sum(CS_energy[1..j])
      "cumulative composter system savings"
      CS_comp_sum[j] = sum(CS_comp[0..j])
   End
```

```
    LCS = sum(PW_comp[0..N])
    //LCS = 0
$ENDIF
```

isentropic pump procedure

Procedure **pump** (h_{in}, P_{in}, P_{out}, F$, η_p : h_{out}, T_{out}, s_{out}, v_{out})

$m := 1$ [kg]

$s_{in} := \mathbf{s}\left(F\$, h = h_{in}, P = P_{in}\right)$

$s_{s,out} := s_{in}$

$h_{s,out} := \mathbf{h}\left(F\$, P = P_{out}, s = s_{s,out}\right)$

$W_{p,s} := m \cdot \left(h_{s,out} - h_{in}\right)$

$W_p := \frac{W_{p,s}}{\eta_p}$

$h_{out} := h_{in} + \frac{W_p}{m}$

$T_{out} := \mathbf{T}\left(F\$, h = h_{out}, P = P_{out}\right)$

$s_{out} := \mathbf{s}\left(F\$, h = h_{out}, P = P_{out}\right)$

$v_{out} := \mathbf{v}\left(F\$, h = h_{out}, P = P_{out}\right)$

End **pump**

SET SYSTEM: Baseline (w/o compost heater) or Modified (w/ compost heater)

SYS$ = 'base'

models - real fluid

W$ = 'water'

A$ = 'air_{ha}'

ambient enviornment

T_{amb} = **ConvertTemp** (F , C – 20) **DESIGN CONDITION Tamb SET: DC_P==1, F_{NG} ==1**

hour = 1

time = 1

P_{amb} = 101325 [Pa]

water tank

$$D_{tank} = 2 \cdot \left| 0.3048 \cdot \frac{m}{ft} \right|$$

$$H_{tank} = 6 \cdot \left| 0.3048 \cdot \frac{m}{ft} \right|$$

$$SA_{tank} = \pi \cdot D_{tank} \cdot H_{tank} + 2 \cdot \pi \cdot \left[\frac{D_{tank}}{2} \right]^2$$

$$Vol_{tank} = \pi \cdot \frac{D_{tank}^{\ 2}}{4} \cdot H_{tank}$$

$h_{conv} = 5$ [W/m^2-K] *convection ht coeff from insulation surface*

$$R'' = 8 \cdot \left| 0.17611016 \cdot \frac{s\text{–}m^2\text{–}K/J}{hr\text{–}ft^2\text{–}F/Btu} \right|$$

$R_{ins} = \frac{R''}{SA_{tank}} \cdot \left| 1 \cdot \frac{K/W}{s\text{–}K/J} \right|$ *water tank with insulation resistance*

$$R_{conv} = \frac{1}{h_{conv} \cdot SA_{tank}}$$

$$R_{tot} = R_{conv} + R_{ins}$$

water tank heat loss

$$Q_{loss,tank} = \left[\frac{\textbf{ConvertTemp}(C, K, T_2) - \textbf{ConvertTemp}(C, K, T_{barn})}{R_{tot}} \right] \cdot 3600 \text{ [s]} \cdot \left| 0.001 \cdot \frac{kJ}{J} \right|$$

pump

$\eta_p = 0.8$ *isentropic pump efficiency*

$PR = 3$ [-] *pump pressure ratio*

floor heat exchanger

$\varepsilon_h = 0.875$ *heat exchanger effectiveness*

composter

$Q_{comp,ton} = 1000 \text{ [BTU/ton]} \cdot \left| 1.055056 \cdot \frac{kJ/ton}{BTU/ton} \right|$ *heat from comp each hour - per ton*

$\rho_{comp} = 640 \text{ [kg/m}^3\text{]} \cdot \left| 0.001102311 \cdot \frac{ton/m^3}{kg/m^3} \right|$ *density of compost*

$Vol_{comp} = \frac{m_{comp}}{\rho_{comp}}$ *mass of compost*

$F_{vol} = 0.1$ *fraction of compost volume to barn volume*

$$Vol_{comp} = Vol_b \cdot \left| 35.31 \cdot \frac{ft^3}{m^3} \right| \cdot F_{vol} \cdot \left| 0.028316847 \cdot \frac{m^3}{ft^3} \right|$$

compost volume scaled to fraction of barn size

$Q_{comp} = Q_{comp,ton} \cdot m_{comp}$ *heat supply per hour from compost*

barn

$T_{barn} = \textbf{ConvertTemp}(F, C, 50)$ **desired barn temperature setpoint**

$$L_b = 40 \cdot \left| 0.3048 \cdot \frac{m}{ft} \right|$$

$$W_b = 30 \cdot \left| 0.3048 \cdot \frac{m}{ft} \right|$$ *barn geometry*

$$H_b = 10 \cdot \left| 0.3048 \cdot \frac{m}{ft} \right|$$

$$Vol_b = L_b \cdot W_b \cdot H_b$$

$$SA_b = 2 \cdot H_b \cdot (L_b + W_b)$$

equivalent resistance from exterior surface

$$R_a = \frac{0.25 \quad [hr\text{-}ft^2\text{-}F/Btu]}{SA_b \cdot \left| 10.76 \cdot \frac{ft^2}{m^2} \right|}$$

corregated steel properties

$k_s = 8$ [Btu/hr-ft-F]

$$L_s = 0.0136 \quad [in] \cdot \left| 0.083333333 \cdot \frac{ft}{in} \right|$$

equivalent conduction resistance through steel walls

$$R_s = \frac{\frac{L_s}{k_s}}{SA_b \cdot \left| 10.76 \cdot \frac{ft^2}{m^2} \right|}$$

equivalend resistance from interior surface

$$R_z = \frac{0.68 \quad [ft^2\text{-}F\text{-}hr/Btu]}{SA_b \cdot \left| 10.76 \cdot \frac{ft^2}{m^2} \right|}$$

$\Delta_{T,f} = \textbf{ConvertTemp}(C, F, T_{barn}) - \textbf{ConvertTemp}(C, F, T_{amb})$ *temperature difference between barn and ambient*

$$\dot{Q}_{loss} = \frac{\Delta_{T,f}}{R_a + R_s + R_z}$$

barn heat loss

$Q_{loss,barn} = \dot{Q}_{loss} \cdot \left| 0.000293071 \cdot \frac{kW}{Btu/hr} \right| \cdot 60 \ [s/min] \cdot 60 \ [min/hr] \cdot 1 \ [hr]$ *actual barn heat loss. can compare to check value*

$Q_{loss,barn,check} = Vol_b \cdot \left| 35.31 \cdot \frac{ft^3}{m^3} \right| \cdot 0.133 \ [BTU/F\text{-}ft^3] \cdot \Delta_{T,f} \cdot \left| 1.055056 \cdot \frac{kJ}{Btu} \right|$ *barn heat loss comparison to eq (1). found in research*

barn heated under dirt

$SA_d = L_b \cdot W_b \cdot \left| 10.76 \cdot \frac{ft^2}{m^2} \right|$

$k_d = 0.7 \cdot \left| 0.57778924 \cdot \frac{Btu/hr\text{–}ft\text{–}F}{W/m\text{–}K} \right|$ *thermal conductivity of soil [0.38-1.7] depending on saturation*

$L_d = 3 \cdot \left| 0.083333333 \cdot \frac{ft}{in} \right|$ *heated piped 4in under soil surface*

equivalent conduction resistance through dirt floor

$R_d = \frac{L_d}{k_d \cdot SA_d}$

dirt floor temp determined from heat rate necessary into barn to account for barn loss

$\dot{Q}_{barn} = \frac{T_{floor} - \textbf{ConvertTemp}(C, F, T_{barn})}{R_d}$

$T_{barn,f} = \textbf{ConvertTemp}(C, F, T_{barn})$

$T_{floor,adj} = \textbf{If}(T_{floor}, T_{barn,f}, T_{barn,f}, T_{barn,f}, T_{floor})$ *adjusted - floor model doesnt get cooled when hot outside - assume stays at minimum barn temp*

cattle heat generation

$N_{cattle} = 20$

$Q_{cattle} = 700 \ [kJ] \cdot N_{cattle}$ *internal heat generation from livestock*

$\dot{Q}_{cattle} = Q_{cattle} \cdot \frac{\left| 0.9478 \cdot \frac{Btu}{kJ} \right|}{1 \ [hr]}$

natural gas heater

$Q_{ng} = 92333 \cdot \left| 1.055056 \cdot \frac{kJ}{Btu} \right|$ *Natural Gas Heater heat per hour: from design condition: -20 F*

mass flow rate

$m_{design} = 596.9 \ [kg]$ *mass flow water @design cond (-20F)*

$DC_p = \frac{m}{m_{design}}$ *pump duty cycle - fraction of time on to deliver mass flow required at design cond*

piping

$$d_{pipe} = 0.75 \cdot \left| 0.0254 \cdot \frac{m}{in} \right|$$

$$A_{pipe} = 3.142 \cdot \left[\frac{d_{pipe}}{2} \right]^2$$

water flow speed check

$$u_{pipe} = \frac{Vol_w}{A_{pipe} \cdot 3600 \ [s]}$$ *check for reasonable flow velocity in piping*

barn zone balance

$$Q_{barn} + Q_{cattle} = Q_{loss,barn}$$ *assume heat to barn + heat from cows == heat leaving barn*

$$\dot{Q}_{barn} + \dot{Q}_{cattle} = \dot{Q}_{loss}$$ *Solve for $\overline{Q}$n heat required to maintain barn temperature*

state 1 - pump outlet

state 1 fixed from pump procedure

$$P_1 = PR \cdot P_3$$

other calculations

$$Vol_w = v_1 \cdot m$$

state 1-2 - tank

energy balance

$$m \cdot h_1 + Q_{ng} \cdot F_{ng} = m \cdot h_2 + Q_{loss,tank}$$

state 2 - tank outlet

$$T_2 = \textbf{ConvertTemp}(F, C, 130)$$ *water outlet temperature setpoint from tank*

$$P_2 = P_1$$

$$h_2 = \textbf{h}(W\$, T = T_2, P = P_2)$$

$$v_2 = \textbf{v}(W\$, T = T_2, P = P_2)$$

$$s_2 = \textbf{s}(W\$, T = T_2, P = P_2)$$

state 2-3 - floor heat exchanger

HX performance

$$\varepsilon_h = \frac{T_2 - T_3}{T_2 - T_{barn}}$$ *solve for T_3 out of HX*

energy balance

$m \cdot h_2 = m \cdot h_3 + Q_{barn}$

state 3 - HX outlet

$P_3 = P_{amb}$

$h_3 = \mathbf{h}(W\$, T = T_3, P = P_3)$

$v_3 = \mathbf{v}(W\$, T = T_3, P = P_3)$

$s_3 = \mathbf{s}(W\$, T = T_3, P = P_3)$

state 3-1 - pump

Call **pump** $(h_3, P_3, P_1, W\$, \eta_p : h_1, T_1, s_1, v_1)$ *determine state 1 properties, after isentropic pump*

energy balance

$m \cdot h_3 + W_p = m \cdot h_1$ *electrical work to drive the pump*

energy used

$E_{ng} = Q_{ng} \cdot F_{ng,adj}$

$E_{elec} = W_p \cdot DC_{p,adj}$

$E_{total} = E_{ng} + E_{elec}$

cost

$F_{ng,adj} = \mathbf{If}(F_{ng}, 0, 0, 0, F_{ng})$ *set natrual gas heater fraction of heat delivered to zero to turn off if negative heat required*

$DC_{p,adj} = \mathbf{If}(DC_p, 0, 0, 0, DC_p)$ *set pump duty cycle to zero if negative heat required from comp and ng*

$ec = 0.12 \ [\$/kW\text{-}hr] \cdot \left| 0.000277778 \cdot \frac{\$/kJ}{\$/kW-hr} \right|$ *electricity price*

$ngc = 0.95 \ [\$/therm] \cdot \left| 0.00000947817 \cdot \frac{\$/kJ}{\$/therm} \right|$ *natural gas price*

$C_{ng} = ngc \cdot Q_{ng} \cdot F_{ng,adj}$ *natural gas cost*

$C_{ec} = ec \cdot W_p \cdot DC_{p,adj}$ *electricity cost*

$COST = C_{ng} + C_{ec}$ *energy costs - not including cost of compost needed for modified system*

SET ==1 TO RUN ECONOMIC ANALYSIS - OTHERWISE SET == 0

ECONOMICS$ = '1'

base system winter cost

$C_{winter,base}$ = **SumParametric** ('WINTER–BASE', 'COST')

modified system winter cost

$C_{winter,mod}$ = **SumParametric** ('WINTER–MOD' , 'COST')

cost difference in one winter heating season

$C_{saved} = C_{winter,base} - C_{winter,mod}$

economics

$N = 10$ *total time loan*

$d = 0.025$ *market discount rate*

$i_{tax} = 0.25$ *effective tax rate*

downpayment and mortgage

$F_{comp} = 7000$ [$] *composter total install (capital) cost*

$f_{dp} = 0.15$ *downpayment fraction*

$C_{dp} = f_{dp} \cdot F_{comp}$ *downpayment cost*

$Mort = (1 - f_{dp}) \cdot F_{comp}$ *total mortgage loan amount*

$m_{rate} = 0.03$ *mortgage interest rate*

fuel

$cc = 25$ [$/ton] *price of materials necessary for compost pile*

$cf = 55$ [$/ton] *resale price per ton of decomposed fertilizer and materials at end of each season*

$labor = 23$ [$/ton] *labor price to install compost each year*

$i_{elec} = 0.03$ *electricity cost inflation rate*

$i_{comp} = 0.03$ *compost price inflation rate*

$i_{fert} = 0.04$ *fertilzer price inflation rate*

initial compost savings

$CS_{comp,0} = - F_{comp} \cdot f_{dp}$

$PW_{comp,0} = CS_{comp,0}$ *present worth of compost savings*

first year fuel/energy costs

$F_{purchase,comp} = 0.5$ *fraction of compost that must be purchased each year*

$F_{compfuel} = cc \cdot m_{comp} \cdot F_{purchase,comp}$ *compost costs first year*

$F_{energy,save} = C_{winter,base} - (C_{winter,mod} + F_{compfuel})$ *fuel/energy savings between base and modified systems (accounting for cost of compost materials needed each year)*

initial compost install labor costs

$CS_{labor,0} = -labor \cdot m_{comp}$

annual mortgage payment

$PR_0 = Mort$

$PR_0 = C_{mortgage} \cdot \mathbf{PWF}(N, 0, m_{rate})$ *determine mortgage cost*

yearly savings

$period_j = j$ (for j = 1 to N)

energy savings costs

$CS_{energy,j} = F_{energy,save} \cdot ((1 + i_{elec})^{(j-1)})$ (for j = 1 to N) *energy savings at year j*

resale

$CS_{resale,j} = cf \cdot m_{comp} \cdot ((1 + i_{fert})^{(j-1)})$ (for j = 1 to N) *resale price of fertilizer at end of each year*

compost install labor costs

$CS_{labor,j} = -labor \cdot m_{comp} \cdot ((1 + d)^{(j-1)})$ (for j = 1 to N) *cost of labor at year j*

interest and principal

$C_{int,j} = m_{rate} \cdot PR_{j-1}$ (for j = 1 to N) *interest at year j*

$PR_j = PR_{j-1} - (C_{mortgage} - C_{int,j})$ (for j = 1 to N) *principal at year j*

income tax

$CS_{tax,j} = C_{int,j} \cdot i_{tax}$ (for j = 1 to N) *income tax savings at year j*

composter savings

$CS_{comp,j} = CS_{energy,j} - C_{mortgage} + CS_{tax,j} + CS_{labor,j} + CS_{resale,j}$ (for j = 1 to N) *yearly system savings*

$PW_{comp,j} = \frac{CS_{comp,j}}{(1 + d)^j}$ (for j = 1 to N) *respective present worth*

cumulative energy savings

$CS_{energy,sum,j} = \mathbf{Sum}(CS_{energy,1..j})$ (for j = 1 to N)

cumulative composter system savings

$CS_{comp,sum,j} = \mathbf{Sum}(CS_{comp,0..j})$ (for j = 1 to N)

$LCS = \mathbf{Sum}(PW_{comp,0..N})$

<u>Uncompiled equations within $IF conditional statements</u>

m*h[1] + Q_comp + (Q_ng*F_ng) = m*h[2] + Q_loss_tank

HF_comp = Q_comp / (Q_comp+(Q_ng*F_ng))"heat fraction: compost heat to toal heat delivered"

m*h[1] + Q_comp + (Q_ng*F_ng) = m*h[2] + Q_loss_tank

HF_comp = Q_comp / (Q_comp+(Q_ng*F_ng))"heat fraction: compost heat to toal heat delivered"